
Abstract
Data collected at the ALS on samples similar to those collected at the GSECARS tomography system are 2.1-2.4 times less noisy, where noise is defined as the standard deviation in the pixel values in nominally uniform areas such as air, glass beads, and water containing a CsI contrast agent.  The lower noise in the ALS images is found whether the data are reconstructed with the ALS software, or with the GSECARS Gridrec software using the default Shepp-Logan filter.  The source of the lower noise is not understood, because a comparison of the statistical variation in the flat-field images shows the ALS data to be 30% noisier than the GSECARS data using that figure of merit.
The GSECARS data can be processed with a Hann filter, rather than the Shepp-Logan filter, to produce images with noise comparable to the ALS data.  But this appears to be at the expense of high-frequency content and does not answer the question as to why the ALS data are less noisy than the GSECARS data when process with the same Shepp-Logan filter.

Introduction

Dorthe Wildenschild has been a frequent user at GSECARS and has collected many tomography data sets at 13-BM-D.  She recently had some data collected on similar samples at the ALS, and noted that the ALS data were easier to segment than the data collected at GSECARS.  In a quick examination of the ALS data it appears to be less noisy than similar data collected at GSECARS.  I decided to make a careful comparison of the data from ALS and GSECARS to see if I can understand the differences.

Data sets

There 2 data sets from the ALS, with properties listed in Table 1.

	Base file name
	Wetballs_9um_2x_binned
	Wetballs_4pt5um_2x

	Pixel size
	9 um
	4.5 um

	NX (horizontal size)
	1451
	2900

	Ny (vertical size)
	125
	250

	Nangles (number of projections)
	720
	720

	Field of view (H, V)
	13.05
	13.05

	Energy
	34.970
	34.970


These data sets were provided both as the raw projections, as well as reconstructed data sets, where the reconstruction was done at the ALS.

Reconstruction
The first comparison is of the GSECARS reconstruction software compared to the ALS reconstructions of the same ALS data sets.  It is possible that the GSECARS reconstruction software is producing noisier reconstructions than the ALS software on the same datasets.  This was tested by reconstructing the ALS data with the GSECARS software as follows:

The ALS data were first converted to the netCDF .volume file used by the GSECARS tomo_display software.  The ALS data consists of dark current files (e.g. Wetballs_9um_2x_binneddrk_0.tif), flat field files (e.g. Wetballs_9um_2x_binnedbak_0.tif), and normal projection files with the sample present (e.g. Wetballs_9um_2x_binned_0.tif) .  The conversion was done with the following IDL procedure:

pro convert_als_tiff, base_file

    ; This program reconstructs the data from the ALS

    ; The tiff files claim to be unsigned 16-bit integers, but they appear 
    ; to have an offset of 32768

    offset = 32768

    ; Get the dimensions of the data

    data = read_tiff(base_file + '_1.tif')

    dims = size(data, /dimensions)

    nx = dims[0]

    ny = dims[1]

    ; Read all the flat fields and average them

    f = findfile(base_file + 'bak*.tif')

    nwhites = n_elements(f)

    whites = uintarr(nx, ny, nwhites)

    for i=0, nwhites-1 do begin

        file = base_file + 'bak_' + strtrim(i, 2) + '.tif'

        data = read_tiff(file) - offset

        whites[0,0,i] = data

    endfor

    ; Correct the flat fields for zingers using double correlation

    for i=0, nwhites-2 do begin

        whites[0,0,i] = remove_tomo_artifacts(whites[*,*,i],   $

                                              image2=whites[*,*,i+1], $

                      /double_correlation, threshold=1.05)

    endfor

    white = total(whites, 3)/nwhites

    ; Read all the dark fields and average them

    f = findfile(base_file + 'drk*.tif')

    ndarks = n_elements(f)

    darks = uintarr(nx, ny, ndarks)

    for i=0, ndarks-1 do begin

        file = base_file + 'drk_' + strtrim(i, 2) + '.tif'

        data = read_tiff(file) - offset

        darks[0,0,i] = data

    endfor

    ; Correct the dark fields for zingers using double correlation

    for i=0, ndarks-2 do begin

        darks[0,0,i] = remove_tomo_artifacts(darks[*,*,i],   $

                                             image2=darks[*,*,i+1], $

                      /double_correlation, threshold=1.05)

    endfor

    dark = total(darks, 3)/ndarks

    ; Subtract the dark field from the flat field

    white = white - dark

    f = findfile(base_file + '_*.tif')

    nprojections = n_elements(f)

    ; The ALS data goes from 0 to 180, but we don't want to 
    ; include the 180 file

    nprojections = nprojections - 1

    vol = intarr(nx, ny, nprojections)

    for i=0, nprojections-1 do begin

        file = base_file + '_' + strtrim(i, 2) + '.tif'

        data = read_tiff(file) - offset

        data = ((float(data) - dark) / white)*10000. + 0.5

        data = remove_tomo_artifacts(data, /zingers, threshold=1.25)

        vol[0,0,i] = fix(data)

    endfor

    write_tomo_volume, base_file + '.volume', vol

end

The above procedure averages all of the dark fields, averages all of the flat fields, removes zingers from the flat fields and projection data, subtracts the dark fields from the flat fields and projections, and normalizes the projections by the flat fields.  It writes out a .volume file that is in the same netCDF format used by the GSECARS tomo_display program.

Once the .volume file was created it was read into tomo_display, and reconstructed.  The center pixel for reconstruction was optimized, and found to be 753.0 for the 9 um data set, and 1505.5 for the 4.5 um data set.

9 micron ALS Data Set
The following IDL program was used to produce the figures and table below for the 9 micron ALS data.

pro print_stats, filter, data

    bead  = data[535:585, 565:605]

    water = data[900:940, 465:485]

    air   = data[605:700, 405:440]

    print, filter 

    print, '  air   = ', mean(air),   '  +-', stddev(air)

    print, '  bead  = ', mean(bead),  '  +-', stddev(bead)

    print, '  water = ', mean(water), '  +-', stddev(water)

end

shepp = read_tomo_volume('Wetballs_9um_2x_binned_Shepp_recon.volume', $

                         zrange=[101, 101])

hann = read_tomo_volume('Wetballs_9um_2x_binned_Hann_recon.volume', $

                        zrange=[101, 101])

hamming = read_tomo_volume('Wetballs_9um_2x_binned_Hamming_recon.volume', $

                            zrange=[101, 101])

als = read_tiff( $

   '../Wetballs_9um_2x_binned_reconstructed/Wetballs_9um_2x_binned_0101.tif')

als = rotate(als, 5)

als = (long(als) - 10855)/3.9

temp = bytscl(als, min=-4000, max=8000)

write_tiff, 'als_recon_slice_101.tif', temp

print_stats, 'Shepp', shepp

print_stats, 'Hann', hann

print_stats, 'Hamming', hamming

print_stats, 'ALS', als

nbins = 1000

hist_min = -4000

hist_max = 8000

x = findgen(nbins)/nbins*(hist_max-hist_min) + hist_min

iplot, x, histogram(als,   min=hist_min, max=hist_max, nbins=nbins), $

       name='ALS', color=[0,0,255], id=id, /ylog, yrange=[1,1e5]

iplot, x, histogram(shepp,   min=hist_min, max=hist_max, nbins=nbins), $

       name='Gridrec Shepp-Logan', color=[255,0,0], overplot=id

iplot, x, histogram(hann,    min=hist_min, max=hist_max, nbins=nbins), $

       name='Gridrec Hann',        color=[0,255,0], overplot=id

;iplot, x, histogram(hamming, min=hist_min, max=hist_max, nbins=nbins), $

;       name='Gridrec Hamming',     color=[0,0,255], overplot=id

;iplot, x, histogram(ramlak,  min=hist_min, max=hist_max, nbins=nbins), $

;       name='Gridrec Ramlak',      color=[0,255,255], overplot=id

end

The following is the reconstruction of slice 101 of the 9um data set using the GSECARS tomo_display software.  This reconstruction was done with the default Shepp-Logan filter.  The reconstructed data had a range of -3821 to 7268.  This was converted to an 8-bit TIFF file using a range of -4000 to +8000, i.e. t = bytscl(data, min=-4000, max=8000)[image: image1.png]



The following is the ALS reconstruction of the same slice.  The ALS data were transformed as follows before creating the image shown below.  An offset of 10855 was subtracted from each pixel, as this was the value outside the image area in the corners.  The data were divided by 3.9, which led to a data range of -2783 to +6431, similar to the range of the tomo_display reconstruction.  The data were converted to an 8-bit TIFF using the same scaling as for the tomo-display image, i.e. -4000 to +8000.

The reconstructions appear to be very similar to the eye.  The GSECARS reconstructions are slightly better in the sense that they have removed some streaks due to zingers that are present in the ALS reconstructions.  These streaks can be seen if the above images are viewed at high magnification in the Word document.

The following is the GSECARS reconstruction with the Hann filter.  On magnification it is considerably smoother than either the ALS or Gridrec/Shepp-Logan reconstructions.

[image: image3.png]



To compare the reconstructions more quantitatively regions of interest were selected in the largest bead, large fluid area, plastic and air in the above images.  The following table summarized the mean and standard deviation on these regions in the two reconstructions.   It can be seen that the GSECARS reconstruction with the Shepp-Logan filter has a slightly higher standard deviation, ~10% larger than the ALS reconstuction.  This difference is almost certainly due to the difference in the filters used in the reconstruction.  The ALS data appear to be reconstructed with a 10’th order Butterworth filter.  For comparison, the mean and standard deviations when the GSECARS Gridrec reconstructions are done with the Hamming and Hann filters are also shown.  Note that the standard deviations are reduced by more than a factor of 2 when using the Hann filter compared to the Shepp-Logan filter.
	
	ALS reconstruction
	GSECARS Gridrec reconstruction

	
	
	Shepp-Logan filter
	Hamming filter
	Hann filter

	Fluid
	4633 +- 263
	4649 +- 290
	4649 +- 151
	4649 +- 141

	Bead
	1688 +- 263
	1570 +- 279
	1569 +- 149
	1569 +- 139

	Air
	61 +- 212
	-61 +- 239
	-61 +- 125
	-61 +- 117


A further comparison of the reconstructions comes from looking at the histograms.  

This plot shows 3 distinct peaks for the air+plastic, beads, and fluid.  The histogram region for the fluid is nearly identical in the ALS and Gridrec/Shepp-Logan reconstructions.  The values for the air are somewhat different, because the ALS reconstruction places 0 in the reconstructed regions with no data (outside the inscribed circle in the reconstructed square) while the GSECARS reconstruction does not do this.  This leads to the sharp spike at 0 in the ALS histogram.  The Gridrec/Hann reconstruction has significantly narrower peaks, and is resolving the air and plastic into separate peaks.

4.5 micron ALS Data Set
The following IDL program was used to produce the figures and table below for the 9 micron ALS data.

; This program compares reconstructions, statistics and histograms for the 4 filter options in Gridrec

pro print_stats, filter, data

    bead  = data[930:1090, 1100:1220]

    water = data[1840:1890, 890:970]

    air   = data[750:870, 1130:1230]

    print, filter 

    print, '  air   = ', mean(air),   '  +-', stddev(air)

    print, '  bead  = ', mean(bead),  '  +-', stddev(bead)

    print, '  water = ', mean(water), '  +-', stddev(water)

end

shepp = read_tomo_volume('Wetballs_4pt5um_2x_Shepp_recon.volume', $

                         zrange=[115, 115])

temp = bytscl(shepp, min=-2000, max=4000)

write_tiff, 'shepp_recon_slice_115.tif', temp

hann = read_tomo_volume('Wetballs_4pt5um_2x_Hann_recon.volume', $

                        zrange=[115, 115])

temp = bytscl(hann, min=-2000, max=4000)

write_tiff, 'hann_recon_slice_115.tif', temp

als = read_tiff( $

    '../Wetballs_4pt5um_2x_reconstructed/Wetballs_4pt5um_2x_0115.tif')

als = rotate(als, 5)

als = (long(als) - 18263)/4.7

temp = bytscl(als, min=-2000, max=4000)

write_tiff, 'als_recon_slice_115.tif', temp

print_stats, 'Shepp', shepp

print_stats, 'Hann', hann

print_stats, 'ALS', als

nbins = 1000

hist_min = -2000

hist_max = 4000

x = findgen(nbins)/nbins*(hist_max-hist_min) + hist_min

iplot, x, histogram(als,   min=hist_min, max=hist_max, nbins=nbins), $

       name='ALS', color=[0,0,255], id=id, /ylog, yrange=[10,1e6]

iplot, x, histogram(shepp,   min=hist_min, max=hist_max, nbins=nbins), $

       name='Gridrec Shepp-Logan', color=[255,0,0], overplot=id

iplot, x, histogram(hann,    min=hist_min, max=hist_max, nbins=nbins), $

       name='Gridrec Hann',        color=[0,255,0], overplot=id

;iplot, x, histogram(hamming, min=hist_min, max=hist_max, nbins=nbins), $

;       name='Gridrec Hamming',     color=[0,0,255], overplot=id

;iplot, x, histogram(ramlak,  min=hist_min, max=hist_max, nbins=nbins), $

;       name='Gridrec Ramlak',      color=[0,255,255], overplot=id

end

The following is the GSECARS reconstruction with Gridrec and the Shepp-Logan filter of slice 115 from the 4.5 um resolution data set.

The following is the ALS reconstruction of the same slice.

The following is the Gridrec/Hann reconstruction of the same slice.

[image: image7.png]



The following is a comparison of the histograms of these reconstructions.
The following is the IDL program that produced the images above and the histogram below.

The histograms have the same pattern at the 9 micron data.  The ALS one is sharper than the Gridrec/Shepp-Logan, but not as sharp as the Gridrec/Hann reconstruction.

To compare the reconstructions more quantitatively regions of interest were again selected in the largest bead, large fluid area, plastic and air in the above images.  The following table summarized the mean and standard deviation on these regions in the two reconstructions.   It can be seen that the GSECARS reconstruction with the Shepp-Logan filter has a significantly higher standard deviation, ~30% larger than the ALS reconstuction.  This difference may just be due to the difference in the filters used in the reconstruction.  For comparison, the mean and standard deviations when the GSECARS Gridrec reconstructions are done with the Hann filter is also shown.  Note that the standard deviations are reduced by more than a factor of 2 when using the Hann filter compared to the Shepp-Logan filter.

	
	ALS reconstruction
	GSECARS tomo_display reconstruction

	
	
	Shepp-Logan filter
	Hann filter

	Fluid
	2281 +- 177
	2285 +- 241
	2285 +- 112

	Bead
	827 +- 178
	796 +- 228
	796 +- 109

	Air
	21 +- 152
	-36 +- 204
	-36 +- 98


GSECARS Data Compared to ALS Data

The next comparison was of data collected at the APS versus that collected at the ALS.  For this comparision we used the following GSECARS data set anfdthe 9 micron ALS data set.
	Base file name
	GSECARS

Nov06\Wildenschild\Loose_bead_MD_1_\Loose_bead_MD_7_1.SPE
	ALS

Wetballs_9um_2x_binned

	Pixel size
	13 um
	9 um

	NX (horizontal size)
	650
	1451

	NY (vertical size)
	427
	125

	NAngles (number of projections)
	720
	720

	Field of view (H, V)
	8.45
	13.05

	Energy
	33.269
	34.970


The first statistic to compare is the noise in the flat fields.  This was computed by averaging all of the flat fields, and then dividing each flat field image by the average.  If everything were ideal the mean of that ratio should be 1.0 for each flat field, with a standard deviation of 0.0.  The statistics for the GSECARS data were computed with the following IDL program.

; Program to compute statistics on the white fields

read_princeton, 'Loose_bead_MD_7_1.SPE', vol

whites = vol[*,*,[0,101,202,303,404,505,606,707,728]]

white_ave = total(whites, 3)/9.

for i=0, 8 do begin

    print, mean(whites[*,*,i]/white_ave), stddev(whites[*,*,i]/white_ave)

endfor

end

This program produced the following output:

IDL> .run white_field_stats   

      1.00014   0.00945157

      1.00048   0.00974791

      1.00223   0.00941428

      1.00024   0.00949259

      1.00128   0.00938722

      1.00060   0.00959688

     0.998515   0.00938402

     0.998114   0.00934515

     0.998404   0.00945103

The means are very close to 1.  The standard deviations are about 0.94%.  This is a good measure of the statistical noise in the measurements.

This same statistic was computed for the 9 um ALS data set using the following IDL program:

pro compute_als_whites, base_file

    ; The tiff files claim to be unsigned 16-bit integers, but they appear to have an

    ; offset of 32768

    offset = 32768

    ; Get the dimensions of the data

    data = read_tiff(base_file + '_1.tif')

    dims = size(data, /dimensions)

    nx = dims[0]

    ny = dims[1]

    ; Read all the flat fields and average them

    f = findfile(base_file + 'bak*.tif')

    nwhites = n_elements(f)

    whites = uintarr(nx, ny, nwhites)

    for i=0, nwhites-1 do begin

        file = base_file + 'bak_' + strtrim(i, 2) + '.tif'

        data = read_tiff(file) - offset

        whites[0,0,i] = data

    endfor

    ; Correct the flat fields for zingers using double correlation

    for i=0, nwhites-2 do begin

        whites[0,0,i] = remove_tomo_artifacts(whites[*,*,i],   $

                                              image2=whites[*,*,i+1], $

                      /double_correlation, threshold=1.05)

    endfor

    white_ave = total(whites, 3)/nwhites

    sdev = fltarr(nwhites)

    ave = fltarr(nwhites)

    for i=0, nwhites-1 do begin

       sdev[i] = stddev(whites[*,*,i]/white_ave)

       ave[i] = mean(whites[*,*,i]/white_ave)

    endfor

    for i=0, nwhites-1 do begin 

        print, ave[i], sdev[i]

    endfor

    print, 'Mean of average, standard deviation:', total(ave)/nwhites, total(sdev)/nwhites

stop

end

IDL> compute_als_whites, 'Wetballs_9um_2x_binned'

      1.00043   0.00937620

     0.996956    0.0176238

      1.00181    0.0125662

      1.00086    0.0118866

      1.00105   0.00738015

     0.995977    0.0158690

     0.998745    0.0231854

      1.00016   0.00807732

      1.00102    0.0115165

      1.00125   0.00688148

      1.00123   0.00769987

     0.999220    0.0146685

     0.999266    0.0120998

     0.998344    0.0137698

      1.00107    0.0123286

      1.00040    0.0112705

     0.999644    0.0131012

      1.00102   0.00954371

     0.998025    0.0107601

      1.00210    0.0185648

      1.00115    0.0175015

      1.00013    0.0128136

     0.999014    0.0115691

     0.999627    0.0109882

      1.00149   0.00728693

Mean of average, standard deviation:      1.00000    0.0123332
The ALS data also has a mean very close to 1.0 for the ratio of each flat field to the average flat field.  However, the range of standard deviations is much larger, because there is significant structure in the flat fields, and this structure is varying with time.  The average standard deviation is 1.2%, which is actually larger than the 0.94% for the GSECARS data.  Most of this is due to the temporal variation of the beam, rather than read noise of the CCD.

The following program was used to computer the table, histogram plot, and images below for the GSECARS data.

pro read_data, filter, data

    data = read_tomo_volume('Loose_bead_MD_7__' + filter + '_recon.volume', zrange=[213, 213])

    write_tiff, 'Loose_bead_MD_7_slice213_' + filter + '.tif', bytscl(data, min=-1000, max=6000)

    m = median(data, 3)

    write_tiff, 'Loose_bead_MD_7_slice213_' + filter + '_median3.tif', bytscl(m, min=-1000, max=6000)

    s = smooth(data, 3)

    write_tiff, 'Loose_bead_MD_7_slice213_' + filter + '_smooth3.tif', bytscl(s, min=-1000, max=6000)

    bead  = data[135:185, 160:210]

    water = data[350:395, 115:140]

    air   = data[370:420, 500:515]

    print, filter

    print, '  air   = ', mean(air),   '  +-', stddev(air)

    print, '  bead  = ', mean(bead),  '  +-', stddev(bead)

    print, '  water = ', mean(water), '  +-', stddev(water)

    print, '  Median 3'

    bead  = m[135:185, 160:210]

    water = m[350:395, 115:140]

    air   = m[370:420, 500:515]

    print, '  air   = ', mean(air),   '  +-', stddev(air)

    print, '  bead  = ', mean(bead),  '  +-', stddev(bead)

    print, '  water = ', mean(water), '  +-', stddev(water)

    print, '  Smooth 3'

    bead  = s[135:185, 160:210]

    water = s[350:395, 115:140]

    air   = s[370:420, 500:515]

    print, '  air   = ', mean(air),   '  +-', stddev(air)

    print, '  bead  = ', mean(bead),  '  +-', stddev(bead)

    print, '  water = ', mean(water), '  +-', stddev(water)

end

read_data, 'Shepp', shepp

read_data, 'Hann', hann

read_data, 'Hamming', hamming

read_data, 'Ramlak', ramlak

als = read_tiff( $

    '../Wetballs_9um_2x_binned_reconstructed/Wetballs_9um_2x_binned_0101.tif')

als = rotate(als, 5)

als = (long(als) - 10855)/3.9

nbins = 1000

hist_min = -2000

hist_max = 8000

x = findgen(nbins)/nbins*(hist_max-hist_min) + hist_min

iplot, x, histogram(shepp,   min=hist_min, max=hist_max, nbins=nbins), name='Shepp-Logan', color=[255,0,0], id=id, yrange=[0,3000]

iplot, x, histogram(hann,    min=hist_min, max=hist_max, nbins=nbins), name='Hann',        color=[0,255,0], overplot=id

iplot, x, histogram(hamming, min=hist_min, max=hist_max, nbins=nbins), name='Hamming',     color=[0,0,255], overplot=id

iplot, x, histogram(ramlak,  min=hist_min, max=hist_max, nbins=nbins), name='Ramlak',      color=[0,255,255], overplot=id

;iplot, x, histogram(als,   min=hist_min, max=hist_max, nbins=nbins)/2., name='ALS', color=[0,0,0], overplot=id

end

~

~

Here is a comparison of the means and standard deviations for the reconstructed values in the phases for the ALS and GSECARS data.  The mean values are different because the ALS normalized to air, and the GSECARS normalized to plastic (there was no air outside the sample).

	
	ALS 9 um data
	GSECARS 13 um data

Gridrec reconstruction

	
	
	Shepp-Logan filter
	Hann filter
	Hamming filter

	Fluid
	4633 +- 263
	4981 +- 606
	4990 +- 310
	4989 +- 330

	Bead
	1688 +- 263
	1405 +- 532
	1403 +- 249
	1403 +- 269

	Air
	61 +- 212
	-504 +- 470
	-506 +- 202
	-506 +- 221


If we compare the standard deviation divided by the mean value for the fluid the ratio is 5.7% for the ALS data, and 12.2% for the GSECARS data with the Shepp-Logan filter.  The ratios for the bead are 15.6% for the ALS data and 37.9% for the GSECARS data.  Thus, the GSECARS reconstuctions have 2.1 to 2.4 times more noise the the ALS data.  This noise is directly reflected in the widths of the histogram peaks, and the relative difficulty in segmentation.  Using the Hann or Hamming filters with Gridrec reduces the noise in the GSECARS data to a level that is very comparable to the ALS data.
The following plot is a histogram of the 4 different filters used in the Gridrec reconstructions above.

The following is the image of slice 213 when reconstructed with the Shepp-Logan filter.

The following is the same slice reconstructed with the Hann filter.  Note that is is significantly less noisy than the Shepp-Logan filter.

The following is the slice reconstructed with the Shepp-Logan filter, and then smoothed with a 3x3 median filter.

The following is the slice reconstructed with the Shepp-Logan filter, and then smoothed with a 3x3 smoothing filter.

Conclusions

The ALS data, even when reconstructed with the normal Gridrec/Shepp-Logan reconstruction is 50% less noisy than the GSECARS data reconstucted in the same way.  The question is why are the GSECARS data is so much more noisy, when the flat fields actually seem to be 30% less noisy than the ALS flat fields?  The only obvious differences between the data are:

· Smaller pixels at ALS (9 microns vs 13 microns)
· More horizontal pixels at the ALS (1451 vs 650)

· More bits in camera (14 (or is it 16?) vs 12)

But the number of camera bits should not really matter, since the noise as measured by the flat field statistics is actually higher.
The GSECARS data can be processed with a Hann filter, rather than the Shepp-Logan filter, to produce images with noise comparable to the ALS data.  But this appears to be at the expense of high-frequency content and does not answer the question as to why the ALS data are less noisy than the GSECARS data when process with the same Shepp-Logan filter.

